

Path1 with matrix input

```
/* pathlmat.sas */
options linesize=79 noovp formdlim='_';
title 'STA2201s06 Path 1 Example with matrix input';

data mat1(type=cov);
  input _type_ $ _name_ $ X1 X2 Y1 Y2;
  datalines;
cov X1 5.1613543 0.4449949 5.1711371 12.1584180
cov X2 0.4449949 9.5089020 0.0029572 28.3694965
cov Y1 5.1711371 0.0029572 7.1990923 14.7572175
cov Y2 12.1584180 28.3694965 14.7572175 118.0597652
n . 300 300 300 300
;
proc print;

proc calis cov vardef=n pcorr;
  /* Analyze the covariance matrix (Default is corr) */
  title2 'Full (unrestricted) Model';
  var X1 X2 Y1 Y2; /* Manifest vars are in the data set */
  lineqs /* Simultaneous equations, separated by commas */
    Y1 = gammall X1 + e1,
    Y2 = beta21 Y1 + gamma22 X2 + e2;
  std /* Variances (not standard deviations) */
    X1 = phi11, /* Optional starting values in parentheses */
    X2 = phi22,
    e1 = psi11,
    e2 = psi22;
  cov /* Covariances */
    X1 X2 = phi12; /* Unmentioned pairs get covariance zero */
  bounds 0.0 < phi11,
    0.0 < phi22,
    0.0 < psi11,
    0.0 < psi22;

proc calis cov vardef=n; /* Analyze the covariance matrix (Default is corr) */
  title2 'Reduced (restricted) Model: gamma22=0';
  var X1 X2 Y1 Y2; /* Manifest vars are in the data set */
  lineqs /* Simultaneous equations, separated by commas */
    Y1 = gammall X1 + e1,
    Y2 = beta21 Y1 + e2;
  std /* Variances (not standard deviations) */
    X1 = phi11, /* Optional starting values in parentheses */
    X2 = phi22,
    e1 = psi11,
    e2 = psi22;
  cov /* Covariances */
    X1 X2 = phi12; /* Unmentioned pairs get covariance zero */
  bounds 0.0 < phi11,
    0.0 < phi22,
    0.0 < psi11,
    0.0 < psi22;

proc iml;
  title2 'Compute G two ways'
```